

Creating your own principal protected security

David G. Messerschmitt

June 2007

Copyright © 2007, David G. Messerschmitt. All rights reserved.

Introduction

Some investors have a higher-than-average aversion to losses in the stock market. An example is a person near or in retirement who fears that a market meltdown such as occurred in 1929 or 1987 could undermine their standard of living in retirement. At the same time, this investor may be concerned about erosion of standard of living through inflation, and wants to protect their real value of their portfolio by owning equities.

For investors like these, the principal protected note (PPN) is an interesting option. Available from many investment banks, they can be purchased from brokers. The idea of the PPN is a guarantee from the bank to return all principal after a period of time, which is typically five to seven years, while also returning a gain that is coupled to the performance of one or more stock market indexes. If a bear stock market, the investor at least recoups principal, and in a bull market the investor participates in those gains. The price paid for the principal protection is less than full "participation" on the upside, meaning that the PPN gain is less than the market gain. The investor may be promised a percentage of the upside gain, or a gain which is capped on a monthly or yearly basis, or some similar promise. Another price is the opportunity cost of getting only principal back; as an alternative, that principal could have been collecting interest in a more conservative investment like bank CD's or Treasury bonds with a guaranteed positive return.

The PPN has some downsides. Typically the full gain is subject to ordinary income tax, and further there are taxes due every year, not just at the maturity of the note. The promises of the note are backed by an investment bank, and the entire investment could be lost if the bank goes out of business. There are also substantial management fees and commissions, and these are not apparent or visible because they are embedded within the promised returns, which the bank hopes to exceed through their trading activity.

An investor can create their own PP security (PPS) by creating a basket of assets that consists of a zero-coupon bond (ZCB), equity index funds participating in the stock market, and index options. At minimum, designing such a security gives the investor a benchmark against which to compare a PPN they may be considering. At maximum, the investor may be able to reap greater returns because management fees are eliminated. Other advantages are that the tax treatment is generally more favorable, and can be optimized by splitting the PPS basket across taxable and tax-deferred accounts, locating each asset where it is treated most favorably. Unlike a PPN, the PPS also does not depend on the creditworthiness of an investment bank. The equity index portion does depend on the creditworthiness of the Options Clearing Corporation (OCC), but it can be arranged so that only the gains (and not the principal protection) depend on the options component of the security. A significant disadvantage of a PPS is that index options are available with a maximum three-year time horizon, so that they must be rolled over at expiration creating another source of risk since the prices of options at the expiration date cannot be predicted. Some investors may view the five- to seven-year guarantees available in a PPN to be attractive, and certainly some of the fees the bank is collecting are justified by the additional risk being assumed at the longer time horizon.

A PPS basket can utilize the following components:

Equity index options expiring on a certain date, typically three years in the future (the longest time horizon available).

- A risk-free investment (such as Treasury Zero Coupon Bond) with a maturity at or near the same expiration date.
- An equity investment keyed to an equity index (such as an index mutual fund or exchanged-traded fund), the same index that underlays the index options.

We discuss two basic approaches to creating a PPS here:

- *Equity proxy.* An options index spread is used to create a leveraged proxy for a stock index that expires worthless if the index has declined at the expiration date. The gain from a ZCB investment balances the cost of the spread position, thus guaranteeing return of principal.
- *Equity hedging.* An equity investment is hedged by an index option that precisely balances any loss in the index. The dividend stream from the equities plus a ZCB component balance the cost of the option position so that the original principal will be recovered if the market declines.

These two forms of PPS are equivalent in terms of return profile. The choice of one or the other thus depends on other practical issues, such as taxes and credit risk.

Disclaimer

Equities and options involve risk and are not suitable for all investors. The strategies discussed in this report are for educational and illustrative purposes only, and should not be construed as an endorsement, recommendation or solicitation to buy or sell securities. Consult with a qualified financial advisor before engaging in equity and option trading. Commissions, taxes and transaction costs are not included in any comparisons. Consult with a qualified tax advisor for the tax implications involved in these strategies.

Review of index options

An index option is an option contract written where the underlying asset is a stock index. We use SPX, the S&P 500 index, as an example. Unlike equity options, where the underlying asset is an actual stock, the index option settlement is in cash rather than shares of stock.

Every option has a premium (the amount of money that passes from buyer to seller when the option contract is established), an exercise date, and a strike price. In the following we will assume the strike price and value are denominated in dollars equal to the value of the index. Money changes hands when an option contract is established (based on the premium) and possibly again when the option contract expires (based on the strike price in relation to the index at that time).

In practice, the SPX contracts both have a \$100 multiplier; that is, they are written in lots of 100 of these options. This makes no difference, except of course when we submit an order to our broker we have to divide the number of options calculated here by \$100. For example, 1000 options contracts purchased as defined here would result in an order for only 10 contracts. Another slight complication is that there exist mini-options whose valuation is based on one-tenth of an index. This is good for the small investor.

Not all brokerage accounts are authorized to trade options. Generally the investor has to apply for this authorization and in the process represent that he or she is knowledgeable about options trading. Creating the simplest form of PPS defined here requires only authorization to buy call or put options. Slightly more sophisticated forms of PPS (those with an upside cap) also require authorization to sell call options.


■ Call option

A call option is the contractual right (but not obligation) to buy the index on the exercise date at the strike price. If we buy one such option contract, then we have a long position in the option. The value of a long position in one contract at the exercise date is denominated in dollars and equals:

```
(* strike=strike price
   index=level of the index at the time of exercise*)
callExerciseValue[strike_, index_] := If[index > strike, index - strike, 0]
```

Let's assume a strike price of 1550 and plot the value of the call option at the expiration date vs the index at the date of expiration.


```
Plot[callExerciseValue[1550, index], {index, 1000, 2000}]
```



This is the cash settlement that the owner of one call option contract will receive at the exercise date in consideration of owning the option contract. Note that our risk in holding the contract is limited to losing the premium we paid, since the contract may be worthless at the expiration date.

We can also sell (or "write") a call option contract, after which we have a short position in the call option. This short position burdens us the obligation to pay (rather than receive) the amount above should the owner of the contract exercise it at the expiration date. Note that there is no "option" on the short side: it is the long side that has the option, but on the short side we are forced to pay up if the buyer demands. We receive payment in the form of a premium as compensation for taking on this obligation. The value of this short position at the exercise data is thus the negative of the value for the long position:

```
Plot[-callExerciseValue[1550, index], {index, 1000, 2000}]
```




The risk we assume in holding a short position in a call option is unlimited. We could lose not only the premium we received, but much more. The reason we might take on this risk is because we are sure that the index will be below the strike price (we are "bearish") in which case the option expires without being exercised and we get to pocket the premium we received. The seller of the call options is assuming much more risk than the buyer, which is why the buyer is compensating the seller through payment of

the premium. The amount that the seller receives is actually slightly less than the buyer pays. The difference is called the "spread", and is what compensates the market makers. In addition, both the buyer and seller will pay a small commission to their broker. In addition, both buyer and sellers deal not directly with one another, but with the OCC. Thus, the creditworthiness that the buyer has to worry about is the OCC, not some anonymous seller.


The premium that the buyer of the call options pays, and the value the seller receives, is a function of the stock price. For the quote above, these two sets of data points are superimposed in the plot below.

```
callPremiums = ListPlot[
  {Transpose[
    {strikes, callBuys}
  ],
  Transpose[
    {strikes, callSells}
  ]},
  AxesLabel → {"Strike price in $", "Premium in $"},
  PlotLabel → "Call options",
  PlotRange → {{0, 2000}, {0, 1000}}
]
```



As the strike price increases, the index is less and less likely to exceed the strike price at the expiration date, so the buyer is less likely to win the bet and the premium is less. The premium is actually made up of two components, the intrinsic value (the value that would be received if the option were exercised immediately after it was bought) and the time value (the value created by the volatility of stock prices). We can illustrate these two components by superimposing the intrinsic value on the previous plot:

```
Show[
Plot[
(* intrinsic value*)
If[quote > strike, quote - strike, 0], {strike, 0, 2000}, PlotRange -> {{0, 2000}, {0, 1000}},
(* average premium*)
callPremiums,
AxesLabel -> {"Strike price in $", "Value in $"},
PlotLabel -> "Call option intrinsic vs total value"
]
```




At low strike prices, the premium is almost all intrinsic value, because the index is statistically unlikely to fall below the strike price at the expiration date. At strike prices above the index at the time of purchase, the call option is said to be "out of the money", meaning that there is zero intrinsic value. The option buyer is paying only for the value that will accrue should the index exceed the strike price at the expiration date, and realizing that benefit declines in probability as the strike price increases so the time value (and the total premium) also decreases.

■ Put option

A put option is the contractual right (but not obligation) to sell the index on the expiration date at the strike price. If we buy one such option contract, then we have a long position in the put option. The value of a long position in one put option contract at the exercise date is denominated in dollars and equals:

```
putExerciseValue[strike_, index_] := If[strike > index, strike - index, 0]
```


```
Plot[putExerciseValue[1550, index], {index, 1000, 2000}]
```



This is the cash settlement that the put option owner will receive at the exercise date in consideration of owning the option contract. Note that our risk in holding the contract is, as with a call option, limited to losing the premium we paid for the option, since the contract may expire worthless.

We can also sell (or "write") a put option contract, in which case we have a short position in the put option. This short position gives us the obligation to pay (rather than receive) the amount above. We receive a premium as compensation for taking on this obligation. The value of this short position at the exercise date is thus the negative of the value for the long position:


```
Plot[-putExerciseValue[1550, index], {index, 1000, 2000}]
```



The risk we assume in holding a short position in a put option is limited to the strike price, but nevertheless substantial. We could lose not only the premium we received, but much more. However, the amount we could lose is limited to the strike price minus the premium we received when we sold the option.

The premiums as a function of strike price are plotted below. We would like to sell an asset at a higher price, and hence a put option sells at a higher price as the strike increases. Saying it another way, as the strike price increases the premium increases because the index at the expiration date is less and less likely to exceed the strike price. This implies that the seller is more likely to lose out, and thus demands a higher premium to compensate for that increasing risk.

```
putPremiums = ListPlot[
  {Transpose[
    {strikes, putBuys}
  ],
  Transpose[
    {strikes, putSells}
  ]},
  AxesLabel → {"Strike price in $", "Premium in $"},
  PlotLabel → "Put options",
  PlotRange → {{0, 2000}, {0, 500}}
]
```



From the foregoing, we see that short positions in an option are generally very risky, and in isolation are considered speculative investments. If the short position is combined in a basket with other elements whose value is correlated with the value of the short position, the basket is not necessarily speculative. In the following, we will be taking short positions in call options, but in every case that short is "covered" by another basket component, eliminating this particular risk.

PPS design

Holding a single option position is not nearly as useful as creating a basket that includes one or more option components. If the basket components have returns that are highly correlated, interesting things can happen. We examine some alternatives, with an emphasis on creating baskets with some form of downside protection, which we call principal protected securities (PPS's).

■ Baseline: leveraged equity basket

Before designing a PPS, it is useful to look at a simpler basket that combines either a long or short position in a risk-free investment (such as a ZCB) with equities. Combining a long position in ZCB reduces the overall risk, and taking a short position increases the risk. The short position is called "buying stock on margin", and entails borrowing money from the broker in order to increase the size of the stock position. Suppose we invest the equity portion in an index fund that mirrors the same index, in this case the S&P 500.

Define a quantity β called the "participation" in stock market gains and losses. It measures the level of risk of the basket. If the overall principal $P = Z + S$ dollars is divided into Z dollars in a risk-free investment over the period of interest (Z may be negative) and S dollars in equities, then the participation β is defined as $\beta = S/P = S/(Z + S)$. It measures the percentage increase in the basket for each percentage increase in stocks. When the ZCB is short, $\beta > 1$, so that the participation is greater than 100%. When the ZCB is long, $\beta < 1$ and the participation is less than 100%. The value V at the end of the period for a given β can be calculated by solving three linear equations. The third equation is the non-trivial one, since it captures how the value changes with capital gains (the first term), dividends (the second term), and the growth in value of the ZCB (the third term). The risk-free interest rate is assumed to be r_F over the period of interest. Note that r_F is the period return, not the annual return. Also, Q is the value of the stock index at the time the basket is purchased.

```
solnStock =
  Solve[
    principal == stock + zcb
    &&
    stock ==  $\beta$  principal
    &&
    valueLevered == stock index / q + stock d + zcb (1 + rF) ,
    {stock, zcb, valueLevered}
  ][[1]] // Simplify


{valueLevered  $\rightarrow$   $\frac{\text{principal} (\text{index} \beta + q (1 + rF + (-1 + d) \beta - rF \beta))}{q}$  ,
  zcb  $\rightarrow$  principal - principal  $\beta$ , stock  $\rightarrow$  principal  $\beta$ }
```

This is best understood by plotting the value of the basket as a function of the index at the end of the period of interest (when the ZCB matures), first for different values of β all below unity (long position in a ZCB). Relative to $\beta = 0$ (the red line), the slope of the value vs. index curve is reduced, and both the upside gains and downside losses are reduced for a given change in the index.

```

valStock = valueLevered /. solnStock /.
{
  rF → Exp[rr period] - 1,
  d → dividendReturn[dd, rr, period],
  q → quote,
  principal → 1000
};
stockPlot = Plot[
  valStock /.  $\beta$  → 1,
  {index, 0, 3000},
  PlotStyle → {Red, Thick}];
Plot[
  Table[valStock, { $\beta$ , 0, .9, .1}],
  {index, 0, 3000},
  PlotLabel → "Risk-reduced $1000 stock basket",
  AxesLabel → {"Index", "Value in $"}
];
Show[stockPlot, %]

```




Turning this around, the value is plotted for a set of β 's greater than unity (short position in the ZCB, or buying the equities on margin). Here the slope is increased, and in effect both upside and downside movements in the index are magnified. Generally those who buy equities on margin are highly confident about upward movements.

```

Plot[
  Table[valStock, { $\beta$ , 1.1, 1.5, .1}],
  {index, 0, 3000},
  PlotLabel → "Margined $1000 stock basket",
  AxesLabel → {"Index", "Value in $"}
];
Show[stockPlot, %]

```




The final step in creating a PPS is to add index options to this basket.

Principal-protected baskets

Leveraging equities with a ZCB is a rather crude tool because it treats the downside and the upside equally. Generally an investor likes upside movements but abhors downside movements. It therefore makes sense to treat them differently. That is possible by adding index options to the mix. We will therefore create PPS's which have a cap (or maximum) on both the upside and downside returns. A cap on the downside has an obvious benefit, but why a cap on the upside? The purpose of this upside cap, which is definitely optional, is to make the participation rate more attractive. A downside cap is a cost in creating the basket, and that cost tends to reduce the participation rate. We can think of that cost as an insurance premium we are paying to limit our downside losses. A cap on the upside yields us a source of money to help pay that premium, and thus increases the participation rate.

A PPS is created with a fixed expiration date in mind, and that expiration date corresponds ideally to the maturity of the ZCB component and the expiration date for the options components. The PPS is designed to achieve a specific value on that expiration date as a function of the unknown value of the index on that date. In the following picture, the PPS value vs. index (black) is compared to an equity-only value (red). The PPS design is determined by four parameters (Q is fixed by the market), and in addition is constrained to have a value at maturity when the index has not changed equal to the equity-only basket. The design parameters are a lower and an upper strike price (s_L and s_H), a participation rate between the lower and higher strikes equal to β_1 , and a participation rate β_2 at higher values of the index. In each case, the participation rate is defined to be unity when the slope of the PPS value and the equity value are equal.

```
Import["Figures/slide1.png", ImageSize -> Medium]
```



In terms of performance, there are four parameters of interest. The two participation rates make up the first two. A third is the downside cap on value decrease, defined as s_L/Q . When $s_L \approx 0$, the PPS is truly principal protected. The fourth is the upside "cap", defined as s_H/Q . When $\beta_2 = \beta_1$, a degenerate case, two of these parameters disappear (the upside cap and the second participation rate). Generally we are interested in $\beta_2 \leq \beta_1$, since it makes sense to reserve the higher participation rate for the range of indexes that is more likely to occur.


The available strike prices put a limit on the range of returns. As shown in the following plots, the maximum range in annualized return is approximately minus 37% to plus 10% (relative to Q) based on the available strike prices and the quoted value of Q . What is plotted is the annualized return for both the index and the caps on PPS value.

```
(* arguments are annualized loss in %, a negative number,
and annualized gain, a positive number *)
sLH[loss_, gain_] :=
(
  periodLoss = Exp[loss percent period] - 1;
  periodGain = Exp[gain percent period] - 1;
  sLi = strikeNear[quote (1 + periodLoss)];
  sHi = strikeNear[quote (1 + periodGain)];
  (* output is an array containing strikes and their indexes *)
  {{sLi, strikes[[sLi]]}, {sHi, strikes[[sHi]]}}
)
```

```

Plot[
  {
    100 returnAnnualized[
      quote,
      sLH[ret, 20][[1]][[2]],
      period
    ],
  },
  {ret, -50, 20},
  PlotPoints → 100,
  PlotRange → {-40, 15}
];
Plot[
  100 returnAnnualized[
    quote,
    sLH[20, ret][[2]][[2]],
    period
  ],
  {ret, -50, 20},
  PlotPoints → 100,
  PlotRange → {-40, 15}
];
Show[%, %%,
  AxesLabel → {"Index %", "Value %"},
  PlotLabel → "Available downside and upside caps in annualized %"]

```



Let's choose some numerical values for the downside and upside caps in the subsequent examples. Let's start with the principal protection case, where the return is not allowed to fall below 0% and the upside return is not allowed to go above 10% annualized.

```

(* cap on downside and upside in annualized % *)
lossCap = 0;
gainCap = 10;

```


We will now demonstrate two methods of achieving the PPS value shown. Since they both yield identical value vs. index profiles at the expiration date, they are equivalent with respect to performance. In terms of practical issues like credit risk and taxes, they

are considerably different.

■ Proxy PPS

A proxy PPS uses index options as a proxy for equity returns. It is constructed out of three components. The largest part of the value at expiration is the ZCB, which is basically the principal guarantee. A long position in a call option gives equity-like returns in the range from the lower strike to the upper strike, and the number of these options determines β_1 . A short position in a call option at the higher strike yields a selling premium that subsidizes the cost of the long call position. The number of short calls determines β_2 .

```
Import["Figures/slide2.png", ImageSize -> Medium]
```



Saying it another way, the long position in the call is a proxy for equities, a risky investment in isolation since it could expire worthless. The ZCB long position balances that risk. The short position in a call is by itself a very speculative investment, since it could lose an unlimited amount of money, but it is balanced by the long call position. Whatever the short call loses, the long call more than makes up for.

Suppose the total principal of P dollars is invested in Z dollars for the ZCB and n_L call options at a lower strike price s_L . In addition, a short position is added in n_H call options at a higher strike price s_H . Suppose further that the cost of one call option at s_L is p_L , and the revenue from selling one call option at s_H is p_H . Then the total cost and revenue must add up to the principal,

$$P = Z + n_L p_L - n_H p_H .$$

The total principal is invested according to the solution to a set of four linear equations. The fourth equation forces the value of the PPS to equal that of an equity basket (including dividends) when the index doesn't change. That is, the value vs index for the two baskets cross one another at Q .

```

Clear[sL, sH, rF, principal, beta1, beta2, nL, nH, pL, pH, zcb];
solnProxy = Solve[
  (* principal is allocated between zcb and options *)
  principal == zcb + nL pL - nH pH
  &&
  (* match number of options to slope between low and high strike*)
  nL == beta1 principal / q
  &&
  (* ditto above high strike *)
  nL - nH == beta2 principal / q
  &&
  (* force value to equal that of the stock portfolio when the index doesn't budge *)
  zcb (1 + rF) + nL (q - sL) == principal (1 + d),
  {zcb, nL, nH, beta1}
] [[1]] // Simplify;
% // TraditionalForm

```

$$\left\{ \begin{array}{l} \beta_1 \rightarrow \frac{q(d - rF) + \beta_2 p_H (rF + 1)}{rF p_H + p_H - p_L + q - p_L rF - s_L}, n_H \rightarrow \frac{\text{principal} (q(d - rF) + \beta_2 (rF p_L + p_L - q + s_L))}{q (rF p_H + p_H - p_L + q - p_L rF - s_L)}, \\ zcb \rightarrow \frac{\text{principal} (q(-d p_L - p_L + q - s_L) + p_H ((-\beta_2 + d + 1) q + \beta_2 s_L))}{q (rF p_H + p_H - p_L + q - p_L rF - s_L)}, n_L \rightarrow \frac{\text{principal} (q(d - rF) + \beta_2 p_H (rF + 1))}{q (rF p_H + p_H - p_L + q - p_L rF - s_L)} \end{array} \right\}$$

Here is the value of the basket at expiration:

```
valueProxy = zcb (1 + rF) + nL callExerciseValue[sL, index] - nH callExerciseValue[sH, index];
```

Let's illustrate this with an example. First, let's fix some standard parameters:

```

Clear[rF, d, q, pL, pH, sL, sH];
ss = sLH[lossCap, gainCap];
substProxy = {
  rF -> Exp[rr period] - 1,
  d -> dividendReturn[dd, rr, period],
  q -> quote,
  pL -> callBuys[[ss[[1]][[1]]]],
  pH -> callSells[[ss[[2]][[1]]]],
  sL -> ss[[1]][[2]],
  sH -> ss[[2]][[2]]
}
{rF -> 0.129075, d -> 0.0459054, q -> 1537.59, pL -> 208.2, pH -> 29.6, sL -> 1550, sH -> 2000}


```

A plot of β_1 vs β_2 reveals a couple of things. First, the highest participation rate is only about 60%, and this occurs when the upside cap is "hard". Second, taking away the upside cap reduces this participation rate.

```

(* we want to limit beta2 ≤ beta1 *)
beta2max = beta1 /. solnProxy /. substProxy /. beta2 → 0;
Plot[
  100 beta1 /. solnProxy /. substProxy,
  {beta2, 0, beta2max},
  PlotRange → {0, Full},
  AxesLabel → {"β2", "β1"},
  PlotLabel → "Participation"
]

```





We can also plot the value of the basket gain, both in raw dollars and in annualized percentage. It behaves as expected. Also shown is the return on a stock basket. The reduction in risk on the downside and the cap on gains on the upside are clearly evident, as is the modest reduction in slope of gain vs. index. The gap between the stock basket and the PPS in the range between strike prices is because we constrained the PPS to preserve principal when the index doesn't change.

```

temp = valueProxy / principal /. solnProxy /. substProxy // Simplify;
Plot[valStock / 1000 /.  $\beta \rightarrow 1$ ,
  {index, 0, 3000},
  PlotStyle -> {Thick, Red}];
valueProxyPlot = Plot[temp /. beta2 -> 0, {index, 0, 3000}];
Show[%%, %, AxesLabel -> {"Index", "Value $"}];
Plot[
  100 returnAnnualized[1000, valStock, period] /. { $\beta \rightarrow 1$ , index -> quote Exp[rp percent period]},
  {rp, -20, 20},
  PlotStyle -> {Thick, Red}];
Plot[
  100 returnAnnualized[1, temp, period] /. {beta2 -> 0, index -> quote Exp[rp percent period]},
  {rp, -20, 20}
];
Show[%%, %, AxesLabel -> {"Index %", "Return %"}];
GraphicsColumn[{{%%%, %}]

```




The participation rate we have shown is fairly low, and this is a bit discouraging. However, we will show shortly that it can be improved considerably by giving up something on the downside. But first, let's show how to design an equivalent hedging PPS.

■ Hedging PPS

A hedging PPS is constructed out of four components. The largest part of the value at expiration is a direct investment in equities. This is, of course, the component that gives equity-like returns. A long position in a put option acts as a hedge when the index drops, since its value at expiration increases even as the value of equities is decreasing. A short position in a call option at the higher strike yields a selling premium that subsidizes the cost of the long put position. The number of short calls determines β_2 . The dividends from the equity position and the maturing ZCB gives a relatively small fixed component that can adjust the value up or down (the latter occurs when the ZCB position is short). Between the two strike prices, the value is determined mainly by the equity basket, and thus it is the equity basket alone that determines β_1 .

```
Import["Figures/slide3.png", ImageSize -> Medium]
```



The size of the four components can be determined by solving five simultaneous linear equations.

```
Clear[sL, sH, rF, principal, beta1, beta2, nL, nH, pL, pH, zcb];
solnHedge = Solve[
  (* principal is allocated between zcb, stock, and options *)
  principal == zcb + stock + nL pL - nH pH
  &&
  (* choose number of put options to hedge losses in the stock portion*)
  nL == stock / q
  &&
  (* choose number of call options shorted to achieve desired beta2 above the high
  strike; do this by setting the net slope equal to the stock basket slope *)
  stock / q - nH == beta2 principal / q
  &&
  (* force basket value to equal the stock basket when
  the index doesn't budge by adjusting zcb and also using dividends *)
  zcb (1 + rF) + stock d + stock == principal (1 + d)
  &&
  (* beta1 is not directly affected by the options *)
  beta1 principal == stock,
  {zcb, stock, nL, nH, beta1}
] [[1]] // Simplify;
% // TraditionalForm
```

$$\left\{ \begin{array}{l} \beta_1 \rightarrow \frac{q(d - rF) + \beta_2 p_H (rF + 1)}{rF p_H + p_H - p_L + d q - p_L rF - q rF}, \\ nL \rightarrow \frac{\text{principal} (q(d - rF) + \beta_2 p_H (rF + 1))}{q(rF p_H + p_H - p_L + d q - p_L rF - q rF)}, nH \rightarrow \frac{\text{principal} (q(d - rF) + \beta_2 (rF p_L + p_L - d q + q rF))}{q(rF p_H + p_H - p_L + d q - p_L rF - q rF)}, \\ zcb \rightarrow -\frac{(d + 1)((\beta_2 - 1) p_H + p_L) \text{principal}}{rF p_H + p_H - p_L + d q - p_L rF - q rF}, \text{stock} \rightarrow \frac{\text{principal} (q(d - rF) + \beta_2 p_H (rF + 1))}{rF p_H + p_H - p_L + d q - p_L rF - q rF} \end{array} \right\}$$

The value of the PPS at the end of the expiration period is:

```
valueHedge = zcb (1 + rF) + stock d + stock index / q +
  nL putExerciseValue[sL, index] - nH callExerciseValue[sH, index];
```

Let's create a data structure with values to substitute:

```

Clear[rF, d, q, pL, pH, sL, sH];
ss = sLH[lossCap, gainCap];
substHedge = {
  rF → Exp[rr period] - 1,
  d → dividendReturn[dd, rr, period],
  q → quote,
  pL → putBuys[[ss[[1]][[1]]]],
  pH → callSells[[ss[[2]][[1]]]],
  sL → ss[[1]][[2]],
  sH → ss[[2]][[2]]
};
Print[substHedge];

```


```
{rF → 0.129075, d → 0.0459054, q → 1537.59, pL → 99.4, pH → 29.6, sL → 1550, sH → 2000}
```

Plotting the participation between strike prices against the participation at indexes higher than the higher strike price, we see similar behavior to the proxy basket. The maximum participation is a bit higher, presumably because we are making slightly different assumptions about interest rates and dividend rates than the market.

```

(* keep beta2 less than the maximum beta1 *)
beta2max = beta1 /. solnHedge /. substHedge /. beta2 → 0;
Plot[
  100 beta1 /. solnHedge /. substHedge,
  {beta2, 0, beta2max},
  PlotRange → {0, Full},
  AxesLabel → {" $\beta_2$ ", " $\beta_1$ "},
  PlotLabel → "Participation"
]

```





Here are plots of the value and return for the hedged basket, which are similar to the proxy basket.

```


temp = valueHedge / principal /. solnHedge /. substHedge // Simplify;
Plot[valStock / 1000 /.  $\beta \rightarrow 1$ ,
  {index, 0, 3000},
  PlotStyle -> {Thick, Red}];
valueHedgePlot = Plot[temp /. beta2 -> 0, {index, 0, 3000}];
Show[%%, %, AxesLabel -> {"Index", "Value $"}];
Plot[
  100 returnAnnualized[1000, valStock, period] /. { $\beta \rightarrow 1$ , index -> quote Exp[rp percent period]},
  {rp, -20, 20},
  PlotStyle -> {Thick, Red}];
Plot[
  100 returnAnnualized[1, temp, period] /. {beta2 -> 0, index -> quote Exp[rp percent period]},
  {rp, -20, 20}
];
Show[%%, %, AxesLabel -> {"Index %", "Return %"}];
GraphicsColumn[{{%%%, %}}]

```



A plot of the value of the two PPS's together shows their strong similarity. Again, the differences are presumably due to differences in the assumed dividend rate and risk-free interest rate between our values and the market pricing of the options.

```
Show[
  valueProxyPlot,
  valueHedgePlot,
  PlotRange -> {.9, 1.3},
  AxesLabel -> {"Index", "Value $"}
]
```



■ Protection vs. participation trade-offs




Let's explore the relationship between downside protection and participation. Since the proxy and hedging PPS's are equivalent, let's deal with the proxy. First, we plot below both the value and return for a PPS with different levels of downside protection, ranging from 0% to 10% annualized, and with an upside cap of 10% annualized. Note that both the downside protection and the upside cap are defined relative to changes in the underlying index, not for the PPS itself. Due to different participation rates, both the downside and upside caps on the PPS differ somewhat from the index downside and upside. This can be seen most clearly for the upside cap, which we have kept fixed for the index but actually varies somewhat for the PPS. Note also that the participation increases as we give up more downside protection. All this is for $\beta_2 = 0$, so we can increase the participation above the higher strike at the expense of lowering the participation between the two strikes. The participation rates are plotted in the third figure. They show that the most important participation rate, that below the upside cap, can be increased in two ways. (a) The upside cap can be made more stringent (β_2 made smaller), or (b) the downside cap relaxed. In particular, it makes sense to choose a downside cap at least as large as the dividend rate, which allows the PPS return to be zero.

```
(* keep the gain cap constant*)
gainCap = 20;
(* other fixed values *)
subst = {
  rF -> Exp[rr period] - 1,
  d -> dividendReturn[dd, rr, period],
  q -> quote
};
```

```

(* plot stock value and return *)
valStockPlot =
  Plot[valStock / 1000 /.  $\beta \rightarrow 1$ ,
    {index, 0, 3000},
    PlotStyle  $\rightarrow$  {Thick, Red}];
retStockPlot =
  Plot[
    100 returnAnnualized[1000, valStock, period] /. { $\beta \rightarrow 1$ , index  $\rightarrow$  quote Exp[rp percent period]},
    {rp, -20, 20},
    PlotStyle  $\rightarrow$  {Thick, Red}];
(* create array of PPS values and returns for different downside caps *)
nn = 10; (* number of caps other than zero % *)
val = Table[0, {nn + 1}];
ret = Table[0, {nn + 1}];
par = Table[0, {nn + 1}];
Do[
  ss = sLH[-(k - 1), gainCap]; (* 1% granularity *)
  substProxy =
    Join[
      subst,
      {
        pL  $\rightarrow$  callBuys[[ss[[1]][[1]]]],
        pH  $\rightarrow$  callSells[[ss[[2]][[1]]]],
        sL  $\rightarrow$  ss[[1]][[2]],
        sH  $\rightarrow$  ss[[2]][[2]]
      }
    ];
  val[[k]] =
    valueProxy/principal /. solnProxy /. substProxy // Simplify;
  ret[[k]] =
    100 returnAnnualized[1, val[[k]], period] /. index  $\rightarrow$  quote Exp[rp percent period];
  par[[k]] = beta1 /. solnProxy /. substProxy // Simplify;
  , {k, 1, nn + 1}
];
Plot[
  val /. beta2  $\rightarrow$  0,
  {index, 0, 3000}
];
Show[valStockPlot, %, AxesLabel  $\rightarrow$  {"Index", "Value $"}];
Plot[
  ret /. beta2  $\rightarrow$  0,
  {rp, -20, 20}
];
Show[retStockPlot, %, AxesLabel  $\rightarrow$  {"Index %", "Return %"}];
Plot[
  par,
  {beta2, 0, 1},
  PlotRange  $\rightarrow$  {0, 1},
  AxesLabel  $\rightarrow$  { $\beta_2$ ,  $\beta_1$ }
];
GraphicsColumn[{"%%%", "%", %}]

```



For many investors, it may make sense to lock an equity basket into a fairly tight range, say plus and minus 10% annualized return. Note that it is the period return, and not the annual return, that is capped, so this approach does not rule out benefiting from larger annual returns or getting hurt by larger annual losses. With this approach, close to full participation in the market upside is possible, over this range.

Comparison

Although the proxy and hedging approaches are equivalent in terms of valuation at the expiration, there are considerable practical differences.

Option pricing. The equivalence of the two approaches depends on the fair pricing of options. Index options are very heavily traded, so substantial dislocations are unlikely. If the two approaches were to differ, an arbitrage opportunity would result.

Taxes. The tax treatment of zero-coupon bonds, equities, and options are quite different. Further, options come under different rules when used in different circumstances, such as "naked", in a "straddle", and when used for hedging. Cash-settlement index options are also treated differently from options on individual stocks. Due to these complexities, a tax advisor should be consulted.

Credit risk. The whole point of a PPS is to protect principal, and a default on any component of a PPS runs directly counter to this. Zero-coupon bonds, if they are US Treasuries, are backed by the full faith and credit of the US Government. Credit problems with an index fund investing in equities are unlikely. The creditworthiness of the OCC is undoubtedly the greatest risk. Fortunately, it has an AAA rating from Standard and Poor's. The investor can compare the two forms of PPS under a worst-case scenario where the OCC defaults on its obligation to option buyers. The investor should also consider whether failure of the OCC is possible, or whether it would be bailed out by the Government given its importance to the financial derivatives industry.

Other reading

For a more detailed introduction to options, spreads, and collars, see [1]. Hedging strategies, and especially their tax implications and match to investor needs are described in [2]. Several industry web sites provide a wealth of educational material [3,4,5]. See [6] for the detailed tax rules, with [7] a less cryptic summary. Information about the financial stability of OCC can be found at [4] and [11]. An excellent series of white papers from the CBOE in [8,9,10] describes options strategies for different circumstances including their tax consequences.

References

■ Options basics

[1] Bodie, Zvi; Kane, Alex; Marcus, Alan. *Essentials of Investments*. McGraw-Hill Irwin, Fifth Edition, 2004.

[2] Miller, Mark. "Hedging Strategies for Protecting Appreciation in Securities and Portfolios". *Journal of Financial Planning*, Oct. 2005.

■ Educational web sites

[3] Chicago Board Options Exchange, Web URL = <http://cboe.com/>.

[4] The Options Clearing Corporation, Web URL = <http://www.optionsclearing.com/>.

[5] The Options Industry Council, educational web site at URL = <http://www.888options.com/>.

■ Taxation

[6] Internal Revenue Service. *Publication 550: Investment Income and Expenses*, 2006.

[7] Ernst and Young, "Taxes and Investing: A Guide for the Individual Investor", available at URL = <http://www.cboe.com/Learn->

Center/pdf/TaxesandInvesting.pdf.

[8] CBOE Investor Series No. 1, "Mutual Funds & Listed Options", white paper available at <http://www.cboe.com/-Institutional/pdf/mutualfund2001.pdf>.

[9] CBOE Investor Series No. 3, "Individual Retirement Accounts and Keogh Plans", white paper available at CBOE Investor Series No. 1, "Mutual Funds & Listed Options", white paper available at <http://www.cboe.com/Institutional/pdf/ira4x92001.pdf>.

[10] CBOE Investor Series No. 6, "High-net-worth Investors & Listed Options", white paper available at <http://www.cboe.com/-LearnCenter/pdf/2001-highnetworth.pdf>.

■ Credit risk

[11] Standard and Poor's, "Options Clearing Corporation", RatingsDirect available at URL = http://optionsclearing.com/about/aaa_rating_07.pdf.

Initialization

The following initialize functions used earlier in this report.

■ Pricing data

■ Options

The following quotes were downloaded on June 4, 2007 from the SPOE web site, and are used in the examples to follow. They list the bid and ask prices for SPX options (options where the underlying asset is the S&P 500 stock index) expiring on Dec. 18, 2009. The first column is the strike price, the next two are the premiums for call options, and the final two are premiums for put options. We always pay the higher of the two prices to purchase an option (take a long position in the option), and receive the lower price to sell an option (take a short position).

```
SetDirectory[$HomeDirectory <> "/Documents/Home/Investments/Principal protection"];
FileNames["*.xls"]
{spXDec09.xls}
```

```

oquotes = Import["spxDec09.xls"][[1]];
Grid[
  Join[
    {"-----Option quotes from CBOE-----", SpanFromLeft},
    {"Strike", "Call(sell)", "Call(buy)", "Put(sell)", "Put(buy)"},
    %
  ],
  Frame → All,
  Alignment → "."
]

```

| -----Option quotes from CBOE----- | | | | |
|-----------------------------------|------------|-----------|-----------|----------|
| Strike | Call(sell) | Call(buy) | Put(sell) | Put(buy) |
| 600. | 935.7 | 939.7 | 0.6 | 0.9 |
| 700. | 849.5 | 853.5 | 1.5 | 1.9 |
| 800. | 764.3 | 768.3 | 2.6 | 4.2 |
| 900. | 680.5 | 684.5 | 5.6 | 7.6 |
| 1000. | 598.4 | 602.4 | 10.3 | 13.5 |
| 1100. | 518.5 | 522.5 | 17.6 | 20.8 |
| 1150. | 479.5 | 483.5 | 21.9 | 25.9 |
| 1200. | 441.4 | 445.4 | 27.3 | 31.3 |
| 1250. | 404.1 | 408.1 | 33.6 | 37.6 |
| 1300. | 367.7 | 371.7 | 40.9 | 44.9 |
| 1350. | 332.4 | 336.4 | 49.2 | 53.2 |
| 1400. | 298.3 | 302.3 | 58.7 | 62.7 |
| 1425. | 281.7 | 285.7 | 63.9 | 67.9 |
| 1450. | 265.5 | 269.5 | 69.5 | 73.5 |
| 1500. | 234.1 | 238.1 | 81.7 | 85.7 |
| 1525. | 218.9 | 222.9 | 88.3 | 92.3 |
| 1550. | 204.2 | 208.2 | 95.4 | 99.4 |
| 1600. | 176. | 180. | 110.8 | 114.8 |
| 1700. | 125.3 | 129.3 | 147.3 | 151.3 |
| 1800. | 83.8 | 87.8 | 192.9 | 196.9 |
| 1900. | 52. | 56. | 248.4 | 252.4 |
| 1950. | 39.7 | 43.7 | 279.7 | 283.7 |
| 2000. | 29.6 | 33.6 | 313.2 | 317.2 |

The period until option expiration is about 2.54 years. We have chosen this because it is the longest time to expiration available as of the date of the quotes.

```

(* dates associated with the data above *)
today = "June 4, 2007";
end = "Dec. 18, 2009";
(* time horizon for investments in years *) period = DateDifference[today, end, "Year"][[1]];
Print[%, " years"];

```

2.53973 years

■ Interest and dividend rates

The following rate data was obtained from other sources, also on June 4, 2007.

```

percent = 0.01;
rr = 4.78percent; (* Risk free interest rate at 3 yr horizon. We have
  actually used 5 year treasury strips since three year strips are not quoted *)
dd = 1.7percent; (* dividend rate on the S&P 500,
  obtained from an exchange traded fund dealing investing in this index *)
quote = 1537.59; (* today's quote for the S&P 500 index *)

```

■ Manipulating option pricing data

It is convenient to break out the options data into row vectors. For example:

```

strikes = IntegerPart[Transpose[oquotes][[1]]]

{600, 700, 800, 900, 1000, 1100, 1150, 1200, 1250, 1300, 1350,
 1400, 1425, 1450, 1500, 1525, 1550, 1600, 1700, 1800, 1900, 1950, 2000}

```

Let's do the same for the prices, and define some convenient functions that manipulate the database.

```

callSells = Transpose[oquotes][[2]];
callBuys = Transpose[oquotes][[3]];
putSells = Transpose[oquotes][[4]];
putBuys = Transpose[oquotes][[5]];

```

We often need to find a strike that approximates a given price.

```

(* function to returns the index of the strike nearest a given price_ *)
strikeNear[price_] :=
Module[{pos, out},
  (* add price to list, sort, and then extract its position *)
  pos = Position[
    Sort[
      Join[strikes, {price}]
    ],
    price
  ][[1]][[1]];
  (* this gives us two candidates for the nearest strike,
  assume the list strikes[[]] is ordered *)
  out = Switch[
    pos,
    (* choose smallest strike *)
    1, 1,
    (* choose largest strike *)
    Length[strikes] + 1, Length[strikes],
    (* choose nearest strike *)
    _, If[
      price - strikes[[pos - 1]] < strikes[[pos]] - price,
      pos - 1,
      pos
    ]
  ];
  out
];

```

■ Period and annualized returns

We will base period returns on a continuous compounding model for simplicity. Under this model, the value grows in proportion to e^{rt} where r is the rate of return (fraction of the underlying asset added to the value each period, which we take to be a year).

```
(* annualized return for a starting and ending value and period *)
Quiet[
  Solve[Exp[r1 p] == v2 / v1, r1][[1]]
] // TraditionalForm


$$\left\{ r1 \rightarrow \frac{\log\left(\frac{v2}{v1}\right)}{p} \right\}$$

```

This leads us to define a function that, given a starting and ending value and period length, returns the annualized return.

```
(* if time is in years, this returns the equivalent continuous-time annualized return *)
returnAnnualized[valueInitial_, valueFinal_, time_] :=
  Log[N[valueFinal / valueInitial]] / time;
```

For example, if the asset doubles in value over 2.54 years, e.g. grows from \$1 to \$2, then the annualized rate of return is about 27.3%.

```
returnAnnualized[1, 2, period];
Print["Annualized return = ", 100 %, "%"];
```

Annualized return = 27.2922%

The accumulation of dividends when holding stocks is an issue. For simplicity assume that dividends accumulate at a constant continuous rate, and the accumulated value also grows at the risk free rate r . This is modeled by the following differential equation:

```
(* v[t]=value of accumulated dividends vs. time*)
v[t] /. DSolve[
  v'[t] == interestRate v[t] + dividendRate principal
  &&
  v[0] == 0,
  v[t],
  t
][[1]] // Simplify


$$\frac{\text{dividendRate} (-1 + e^{\text{interestRate} t}) \text{principal}}{\text{interestRate}}$$

```

This leads us to define an accumulated dividend period return function:

```
dividendReturn[dividend_, return_, time_] := dividend (Exp[return time] - 1) / return;
```

For example, if we were to hold a basket of stocks with fixed dividend rate dd for 2.54 years, then the total accumulated dividends are approximately 4.6% of the initial principal, and the annualized return is approximately 1.77% (higher than the dividend rate d because of compounding).

```
dividendReturn[dd, rr, period];
returnAnnualized[1, 1 + %, period];
Print["Period return = ", 100 %%, "%", " Annualized return = ", 100 %, "%"]
```

Period return = 4.59054% Annualized return = 1.76724%